

NAG Fortran Library Routine Document

E01SGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E01SGF generates a two-dimensional interpolant to a set of scattered data points, using a modified Shepard method.

2 Specification

```
SUBROUTINE E01SGF(M, X, Y, F, NW, NQ, IQ, LIQ, RQ, LRQ, IFAIL)
INTEGER          M, NW, NQ, IQ(LIQ), LIQ, LRQ, IFAIL
real           X(M), Y(M), F(M), RQ(LRQ)
```

3 Description

This routine constructs a smooth function $Q(x, y)$ which interpolates a set of m scattered data points (x_r, y_r, f_r) , for $r = 1, 2, \dots, m$, using a modification of Shepard's method. The surface is continuous and has continuous first partial derivatives.

The basic Shepard (1968) method interpolates the input data with the weighted mean

$$Q(x, y) = \frac{\sum_{r=1}^m w_r(x, y) q_r}{\sum_{r=1}^m w_r(x, y)},$$

where

$$q_r = f_r \text{ and } w_r(x, y) = \frac{1}{d_r^2} \text{ and } d_r^2 = (x - x_r)^2 + (y - y_r)^2.$$

The basic method is global in that the interpolated value at any point depends on all the data, but this routine uses a modification (see Franke and Nielson (1980), Renka (1988c)), whereby the method becomes local by adjusting each $w_r(x, y)$ to be zero outside a circle with centre (x_r, y_r) and some radius R_w . Also, to improve the performance of the basic method, each q_r above is replaced by a function $q_r(x, y)$, which is a quadratic fitted by weighted least-squares to data local to (x_r, y_r) and forced to interpolate (x_r, y_r, f_r) . In this context, a point (x, y) is defined to be local to another point if it lies within some distance R_q of it. Computation of these quadratics constitutes the main work done by this routine.

The efficiency of the routine is further enhanced by using a cell method for nearest neighbour searching due to Bentley and Friedman (1979).

The radii R_w and R_q are chosen to be just large enough to include N_w and N_q data points, respectively, for user-supplied constants N_w and N_q . Default values of these parameters are provided by the routine, and advice on alternatives is given in Section 8.2.

This routine is derived from the routine QSHEP2 described by Renka (1988a).

Values of the interpolant $Q(x, y)$ generated by this routine, and its first partial derivatives, can subsequently be evaluated for points in the domain of the data by a call to E01SHF.

4 References

Bentley J L and Friedman J H (1979) Data structures for range searching *ACM Comput. Surv.* **11** 397–409

Franke R and Nielson G (1980) Smooth interpolation of large sets of scattered data *Internat. J. Num. Methods Engrg.* **15** 1691–1704

Renka R J (1988a) Algorithm 660: QSHEP2D: Quadratic Shepard method for bivariate interpolation of scattered data *ACM Trans. Math. Software* **14** 149–150

Renka R J (1988c) Multivariate interpolation of large sets of scattered data *ACM Trans. Math. Software* **14** 139–148

Shepard D (1968) A two-dimensional interpolation function for irregularly spaced data *Proc. 23rd Nat. Conf. ACM* 517–523 Brandon/Systems Press Inc., Princeton

5 Parameters

- 1: M – INTEGER *Input*
On entry: m , the number of data points.
Constraint: $M \geq 6$.

- 2: X(M) – *real* array *Input*
3: Y(M) – *real* array *Input*
On entry: the Cartesian coordinates of the data points (x_r, y_r) , for $r = 1, 2, \dots, m$.
Constraint: these coordinates must be distinct, and must not all be collinear.

- 4: F(M) – *real* array *Input*
On entry: the data values f_r , for $r = 1, 2, \dots, m$.

- 5: NW – INTEGER *Input*
On entry: the number N_w of data points that determines each radius of influence R_w , appearing in the definition of each of the weights w_r , for $r = 1, 2, \dots, m$ (see Section 3). Note that R_w is different for each weight. If $NW \leq 0$ the default value $NW = \min(19, M - 1)$ is used instead.
Constraint: $NW \leq \min(40, M - 1)$.

- 6: NQ – INTEGER *Input*
On entry: the number N_q of data points to be used in the least-squares fit for coefficients defining the nodal functions $q_r(x, y)$ (see Section 3). If $NQ \leq 0$ the default value $NQ = \min(13, M - 1)$ is used instead.
Constraint: $NQ \leq 0$ or $5 \leq NQ \leq \min(40, M - 1)$.

- 7: IQ(LIQ) – INTEGER array *Output*
On exit: integer data defining the interpolant $Q(x, y)$.

- 8: LIQ – INTEGER *Input*
On entry: the dimension of the array IQ as declared in the (sub)program from which E01SGF is called.
Constraint: $LIQ \geq 2 \times M + 1$.

- 9: RQ(LRQ) – *real* array *Output*
On exit: real data defining the interpolant $Q(x, y)$.

- 10: LRQ – INTEGER *Input*
On entry: the dimension of the array RQ as declared in the (sub)program from which E01SGF is called.
Constraint: $LRQ \geq 6 \times M + 5$.
- 11: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $M < 6$,
 or $0 < NQ < 5$,
 or $NQ > \min(40, M - 1)$,
 or $NW > \min(40, M - 1)$,
 or $LIQ < 2 \times M + 1$,
 or $LRQ < 6 \times M + 5$.

IFAIL = 2

On entry, $(X(i), Y(i)) = (X(j), Y(j))$ for some $i \neq j$.

IFAIL = 3

On entry, all the data points are collinear. No unique solution exists.

7 Accuracy

On successful exit, the function generated interpolates the input data exactly and has quadratic accuracy.

8 Further Comments

8.1 Timing

The time taken for a call to E01SGF will depend in general on the distribution of the data points. If X and Y are uniformly randomly distributed, then the time taken should be $O(M)$. At worst $O(M^2)$ time will be required.

8.2 Choice of N_w and N_q

Default values of the parameters N_w and N_q may be selected by calling E01SGF with $NW \leq 0$ and $NQ \leq 0$. These default values may well be satisfactory for many applications.

If non-default values are required they must be supplied to E01SGF through positive values of NW and NQ. Increasing these parameters makes the method less local. This may increase the accuracy of the resulting interpolant at the expense of increased computational cost. The default values

$NW = \min(19, M - 1)$ and $NQ = \min(13, M - 1)$ have been chosen on the basis of experimental results reported in Renka (1988c). In these experiments the error norm was found to vary smoothly with N_w and N_q , generally increasing monotonically and slowly with distance from the optimal pair. The method is not therefore thought to be particularly sensitive to the parameter values. For further advice on the choice of these parameters see Renka (1988c).

9 Example

This program reads in a set of 30 data points and calls E01SGF to construct an interpolating function $Q(x, y)$. It then calls E01SHF to evaluate the interpolant at a set of points.

Note that this example is not typical of a realistic problem: the number of data points would normally be larger.

9.1 Program Text

```
*      E01SGF Example Program Text
*      Mark 18 Release. NAG Copyright 1997.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LIQ, LRQ
      PARAMETER        (MMAX=100,NMAX=100,LIQ=2*MMAX+1,LRQ=6*MMAX+5)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, M, N, NQ, NW
*      .. Local Arrays ..
      real             F(MMAX), Q(NMAX), QX(NMAX), QY(NMAX), RQ(LRQ),
+                     U(NMAX), V(NMAX), X(MMAX), Y(MMAX)
      INTEGER          IQ(LIQ)
*      .. External Subroutines ..
      EXTERNAL         E01SGF, E01SHF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'E01SGF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)

*
*      Input the number of nodes.
*
      READ (NIN,*) M
      IF (M.GT.0 .AND. M.LE.MMAX) THEN
*
*          Input the data points X,Y and F.
*
          DO 20 I = 1, M
              READ (NIN,*) X(I), Y(I), F(I)
20          CONTINUE
*
*          Generate the interpolant.
*
          NQ = 0
          NW = 0
          IFAIL = 0
          CALL E01SGF(M,X,Y,F,NW,NQ,IQ,LIQ,RQ,LRQ,IFAIL)
*
*          Input the number of evaluation points.
*
          READ (NIN,*) N
*
*          Input the evaluation points.
*
          DO 40 I = 1, N
              READ (NIN,*) U(I), V(I)
40          CONTINUE
*
*          Evaluate the interpolant using E01SHF.
```

```

*
      IFAIL = -1
      CALL E01SHF(M,X,Y,F,IQ,LIQ,RQ,LRQ,N,U,V,Q,QX,QY,IFAIL)
*
      WRITE (NOUT,*) '      I      U(I)      V(I)      Q(I)'
      DO 60 I = 1, N
        WRITE (NOUT,99999) I, U(I), V(I), Q(I)
60    CONTINUE
*
      END IF
      STOP
*
99999 FORMAT (1X,I6,3F10.2)
      END

```

9.2 Program Data

E01SGF Example Program Data

30			M, the number of data points
11.16	1.24	22.15	X, Y, F data point definition
12.85	3.06	22.11	
19.85	10.72	7.97	
19.72	1.39	16.83	
15.91	7.74	15.30	
0.00	20.00	34.60	
20.87	20.00	5.74	
3.45	12.78	41.24	
14.26	17.87	10.74	
17.43	3.46	18.60	
22.80	12.39	5.47	
7.58	1.98	29.87	
25.00	11.87	4.40	
0.00	0.00	58.20	
9.66	20.00	4.73	
5.22	14.66	40.36	
17.25	19.57	6.43	
25.00	3.87	8.74	
12.13	10.79	13.71	
22.23	6.21	10.25	
11.52	8.53	15.74	
15.20	0.00	21.60	
7.54	10.69	19.31	
17.32	13.78	12.11	
2.14	15.03	53.10	
0.51	8.37	49.43	
22.69	19.63	3.25	
5.47	17.13	28.63	
21.67	14.36	5.52	
3.31	0.33	44.08	End of data points
5			N, the number of evaluation points
20.00	3.14		U, V evaluation point definition
6.41	15.44		
7.54	10.69		
9.91	18.27		
12.30	9.22		End of evaluation points

9.3 Program Results

E01SGF Example Program Results

I	U(I)	V(I)	Q(I)
1	20.00	3.14	15.89
2	6.41	15.44	34.05
3	7.54	10.69	19.31
4	9.91	18.27	13.68
5	12.30	9.22	14.56